Vergleich von "Gammaverteilung" und Lognormalverteilung

1) exponentiell gedämpfte Potenz bzw. "Gammaverteilung"

Definition und Eigenschaften

■ Definition

In[164]:=
```
Clear[α, x0]
p1[x_] = x^α * Exp[-x / x0] / (x0^(α + 1) * Gamma[α + 1])
```

Out[165]= $\frac{e^{-\frac{x}{x0}} x^{\alpha} x0^{-1-\alpha}}{\text{Gamma}[1+\alpha]}$

■ p1 (x) ist richtig normiert ...

In[166]:=
```
Integrate[p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0]
```

Out[166]= 1

■ Bestimmung des Maximums

In[167]:=
```
p1maximum = x /. Solve[D[p1[x], {x, 1}] == 0, x]
```

Out[167]= {x0 α}

■ Momente allgemein berechnet

In[168]:=
```
Integrate[x * p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0]
Integrate[x^2 * p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0]
Integrate[x^3 * p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0]
Integrate[x^4 * p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0]
Integrate[x^n * p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0 && Re[n] > 1]
```

Out[168]= $x0\,(1+\alpha)$

Out[169]= $x0^2\,(1+\alpha)\,(2+\alpha)$

Out[170]= $\frac{x0^3\,\text{Gamma}[4+\alpha]}{\text{Gamma}[1+\alpha]}$

Out[171]= $\frac{x0^4\,\text{Gamma}[5+\alpha]}{\text{Gamma}[1+\alpha]}$

Out[172]= $\frac{x0^n\,\text{Gamma}[1+n+\alpha]}{\text{Gamma}[1+\alpha]}$

■ Erwartungswert in Abhängigkeit von α und x0

In[173]:=
```
p1mittelwert = x0 * (α + 1)
```

Out[173]= $x0\,(1+\alpha)$

■ Momente in Abhängigkeit von α und x0

In[177]:= $\text{p1xmoment}[n\_] = x0^n \prod_{k=1}^{n} (\alpha + k);$

```
p1xmoment[1]
p1xmoment[2]
p1xmoment[3]
p1xmoment[4]
```

Out[178]= $x0\,(1+\alpha)$

Out[179]= $x0^2\,(1+\alpha)\,(2+\alpha)$

Out[180]= $x0^3\,(1+\alpha)\,(2+\alpha)\,(3+\alpha)$

Out[181]= $x0^4\,(1+\alpha)\,(2+\alpha)\,(3+\alpha)\,(4+\alpha)$

■ allgemeine Bestimmung der Varianz

In[182]:=
```
p1varianz =
  Integrate[(x - p1mittelwert)^2 * p1[x], {x, 0, ∞}, Assumptions → Re[x0] > 0 && Re[α] > 0]
```

Out[182]= $x0^2\,(1+\alpha)$

■ Beispielverteilung

In[183]:=
```
x0 = 2;
α = 2;
Plot[p1[x], {x, 0, 20}, PlotRange → {0, 0.15}, PlotStyle → {Directive[Black, Thick]}]
Clear[α, x0]
```

Out[185]=



■ Wie α, x0, Erwartungswert, Maximum und Varianz voneinander abhaengen..

In[187]:=
```
p1maximum
p1mittelwert
p1varianz
```

Out[187]= $\{x0\,\alpha\}$

Out[188]= $x0\,(1+\alpha)$

Out[189]= $x0^2\,(1+\alpha)$

In[190]:=
```
p1maxi[α_, x0_] = α * x0;
p1mittel[α_, x0_] = x0 (1 + α);
p1vari[α_, x0_] = x0^2 (1 + α);
alphamittelmaxi[mittel_, maxi_] = maxi / (mittel - maxi);
alphamittelvari[mittel_, vari_] = mittel^2 / vari - 1;
x0mittelmaxi[mittel_, maxi_] = mittel - maxi;
x0mittelvari[mittel_, vari_] = vari / mittel;
```

Verteilung in Abhängigkeit von Erwartungswert und Maximum

In[197]:=
```
Clear[xmittel, xmax]
p1mod1alpha = alphamittelmaxi[xmittel, xmax];
p1mod1x0 = x0mittelmaxi[xmittel, xmax];
p1mod1[x_] =
 x^p1mod1alpha * Exp[-x / p1mod1x0] / (p1mod1x0^(p1mod1alpha + 1) * Gamma[p1mod1alpha + 1])
```

Out[200]=

$$\frac{e^{-\frac{x}{-xmax+xmittel}}\, x^{\frac{xmax}{-xmax+xmittel}}\, (-xmax+xmittel)^{-1-\frac{xmax}{-xmax+xmittel}}}{\mathrm{Gamma}\left[1+\frac{xmax}{-xmax+xmittel}\right]}$$

■ selbe Beispielverteilung wie oben

In[201]:=
```
x0 = 2;
α = 2;
xmax = p1maxi[α, x0]
xmittel = p1mittel[α, x0]
Needs["PlotLegends`"]
Plot[{p1mod1[x], p1[x]}, {x, 0, 20}, PlotRange → {0, 0.15},
 PlotStyle → {Directive[Black, Thick], Directive[Orange, Dashed, Thick]},
 PlotLegend → {"Gammavert(xmax(x0,α),xmittel(x0,α))", "Gammavert(x0,α)"},
 LegendPosition → {-0.1, 0.1}, LegendShadow → {0, 0},
 LegendSize → {1.2, 0.3}, LegendTextSpace → 15]
Clear[xmittel, xmax, α, x0]
```

Out[203]= 4

Out[204]= 6

Out[206]=



Verteilung in Abhängigkeit von Erwartungswert und Varianz

In[208]:=
```
Clear[xmittel, xvari]
p1mod2alpha = alphamittelvari[xmittel, xvari];
p1mod2x0 = x0mittelvari[xmittel, xvari];
p1mod2[x_] =
 x^p1mod2alpha * Exp[-x / p1mod2x0] / (p1mod2x0^(p1mod2alpha + 1) * Gamma[p1mod2alpha + 1])
```

Out[211]=

$$\frac{e^{-\frac{x\,\text{xmittel}}{\text{xvari}}}\, x^{-1+\frac{\text{xmittel}^2}{\text{xvari}}} \left(\frac{\text{xvari}}{\text{xmittel}}\right)^{-\frac{\text{xmittel}^2}{\text{xvari}}}}{\text{Gamma}\left[\frac{\text{xmittel}^2}{\text{xvari}}\right]}$$

- **selbe Beispielverteilung wie oben**

In[212]:=
```
x0 = 2;
α = 2;
xvari = p1vari[α, x0]
xmittel = p1mittel[α, x0]
Plot[{p1mod2[x], p1[x]}, {x, 0, 20}, PlotRange → {0, 0.15},
 PlotStyle → {Directive[Black, Thick], Directive[Green, Dashed, Thick]},
 PlotLegend → {"Gammavert(xvari(x0,α),xmittel(x0,α))", "Gammavert(x0,α)"},
 LegendPosition → {-0.1, 0.1}, LegendShadow → {0, 0},
 LegendSize → {1.2, 0.3}, LegendTextSpace → 15]
Clear[xmittel, xmax, xvari, α, x0]
```

Out[214]= 12

Out[215]= 6

Out[216]=


In[218]:=

2) Lognormalverteilung

Definition und Eigenschaften

- **Definition**

In[219]:=
```
p2[x_] = 1 / (x * S * Sqrt[2 * Pi]) * Exp[- (Log[x] - μ) ^2 / (2 * S^2)]
```

Out[219]= $\frac{e^{-\frac{(-\mu+\mathrm{Log}[x])^2}{2 S^2}}}{\sqrt{2\pi}\, S\, x}$

- **richtig normiert ...**

In[220]:=
```
Integrate[p2[x], {x, 0, ∞}, Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
```

Out[220]= 1

Bestimmung des Maximums

In[221]:=
```
p2maximum = x /. Solve[D[p2[x], {x, 1}] == 0, x]
```

Out[221]= $\{e^{-S^2+\mu}\}$

Momente allgemein berechnet

In[222]:=
```
Integrate[x * p2[x], {x, 0, ∞},
 Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
Integrate[x^2 * p2[x], {x, 0, ∞},
 Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
Integrate[x^3 * p2[x], {x, 0, ∞},
 Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
Integrate[x^4 * p2[x], {x, 0, ∞},
 Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
Integrate[x^n * p2[x], {x, 0, ∞},
 Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
```

Out[222]= $e^{\frac{S^2}{2}+\mu}$

Out[223]= $e^{2\,(S^2+\mu)}$

Out[224]= $e^{\frac{9\,S^2}{2}+3\,\mu}$

Out[225]= $e^{8\,S^2+4\,\mu}$

Out[226]= $e^{\frac{n^2\,S^2}{2}+n\,\mu}$

Erwartungswert in Abhängigkeit von S und μ

In[227]:= p2mittelwert = $e^{\frac{S^2}{2}+\mu}$;

allgemeine Bestimmung der Varianz

In[228]:=
```
p2varianz = Integrate[(x - p2mittelwert)^2 * p2[x], {x, 0, ∞},
  Assumptions → Re[S] > 0 && Re[μ] > 0 && Im[S] == 0 && Im[μ] == 0]
```

Out[228]= $e^{S^2+2\,\mu}\left(-1+e^{S^2}\right)$

■ Beispielverteilung

In[229]:=
```
S = 1;
μ = 2;
Plot[p2[x], {x, 0, 20}, PlotRange → {0, 0.11}, PlotStyle → {Directive[Black, Thick]}]
Clear[S, μ]
```

Out[231]=



■ Wie S, μ, Erwartungswert, Maximum und Varianz voneinander abhaengen..

In[233]:=
```
p2maximum
p2mittelwert
p2varianz
```

Out[233]= $\{e^{-S^2+\mu}\}$

Out[234]= $e^{\frac{S^2}{2}+\mu}$

Out[235]= $e^{S^2+2\mu}\left(-1+e^{S^2}\right)$

In[236]:=

$p2maxi[S\_, \mu\_] = e^{-S^2+\mu};$

$p2mittel[S\_, \mu\_] = e^{\frac{S^2}{2}+\mu};$

$p2vari[S\_, \mu\_] = e^{S^2+2\mu}\left(-1+e^{S^2}\right);$

$\mu mittelmaxi[mittel\_, maxi\_] = \frac{1}{3}(Log[maxi] + 2\, Log[mittel]);$

$Smittelmaxi[mittel\_, maxi\_] = \sqrt{\frac{2}{3}}\sqrt{Log\left[\frac{mittel}{maxi}\right]};$

$Smittelvari[mittel\_, vari\_] = \sqrt{Log\left[\frac{mittel^2 + vari}{mittel^2}\right]};$

$\mu mittelvari[mittel\_, vari\_] = Log\left[\frac{mittel^2}{\sqrt{mittel^2 + vari}}\right];$

Verteilung in Abhängigkeit von Erwartungswert und Maximum

In[243]:=
```
Clear[xmittel, xmax]
p2mod1S = Smittelmaxi[xmittel, xmax];
p1mod1μ = μmittelmaxi[xmittel, xmax];
p2mod1[x_] = 1 / (x * p2mod1S * Sqrt[2 * Pi]) * Exp[- (Log[x] - p1mod1μ) ^2 / (2 * p2mod1S ^2)]
```

Out[246]=
$$\frac{e^{-\frac{3\left(\mathrm{Log}[x]+\frac{1}{3}\left(-\mathrm{Log}[xmax]-2\,\mathrm{Log}[xmittel]\right)\right)^2}{4\,\mathrm{Log}\left[\frac{xmittel}{xmax}\right]}}\sqrt{\frac{3}{\pi}}}{2\,x\,\sqrt{\mathrm{Log}\left[\frac{xmittel}{xmax}\right]}}$$

- **selbe Beispielverteilung wie oben**

In[247]:=
```
S = 1;
μ = 2;
xmax = p2maxi[S, μ]
xmittel = p2mittel[S, μ]
Plot[{p2mod1[x], p2[x]}, {x, 0, 20}, PlotRange → {0, 0.11},
 PlotStyle → {Directive[Black, Thick], Directive[Orange, Dashed, Thick]},
 PlotLegend → {"Lognormalvert(xmax(S,μ),xmittel(S,μ))", "Lognormalvert(S,μ)"},
 LegendPosition → {-0.1, 0.1}, LegendShadow → {0, 0},
 LegendSize → {1.1, 0.3}, LegendTextSpace → 15]
Clear[xmittel, xmax, xvari, S, μ]
```

Out[249]= e

Out[250]= $e^{5/2}$

Out[251]=



Verteilung in Abhängigkeit von Erwartungswert und Varianz

In[253]:=
```
Clear[xmittel, xvari]
p2mod2S = Smittelvari[xmittel, xvari];
p1mod2μ = μmittelvari[xmittel, xvari];
p2mod2[x_] = 1 / (x * p2mod2S * Sqrt[2 * Pi]) * Exp[- (Log[x] - p1mod2μ) ^2 / (2 * p2mod2S ^ 2)]
```

Out[256]=
$$\frac{e^{-\frac{\left(\text{Log}[x]-\text{Log}\left[\frac{\text{xmittel}^2}{\sqrt{\text{xmittel}^2+\text{xvari}}}\right]\right)^2}{2\,\text{Log}\left[\frac{\text{xmittel}^2+\text{xvari}}{\text{xmittel}^2}\right]}}}{\sqrt{2\pi}\;x\;\sqrt{\text{Log}\left[\frac{\text{xmittel}^2+\text{xvari}}{\text{xmittel}^2}\right]}}$$

■ selbe Beispielverteilung wie oben

In[257]:=
```
S = 1;
μ = 2;
xvari = p2vari[S, μ]
xmittel = p2mittel[S, μ]
Plot[{p2mod2[x], p2[x]}, {x, 0, 20}, PlotRange → {0, 0.11},
 PlotStyle → {Directive[Black, Thick], Directive[Green, Dashed, Thick]},
 PlotLegend → {"Lognormalvert(xvari(S,μ),xmittel(S,μ))", "Lognormalvert(S,μ)"},
 LegendPosition → {-0.1, 0.1}, LegendShadow → {0, 0},
 LegendSize → {1.1, 0.3}, LegendTextSpace → 15]
Clear[xmittel, xmax, xvari, S, μ]
```

Out[259]= $(-1 + e)\, e^5$

Out[260]= $e^{5/2}$

Out[261]=



In[263]:=

3) Vergleich der beiden Verteilungen

gleiche Varianz und gleicher Erwartungswert

■ Verlauf

In[348]:=
```
xvari = 10;
xmittel = 6;
Plot[{p1mod2[x], p2mod2[x]}, {x, 0, 30},
 PlotStyle → {Directive[Black, Thick], Directive[Green, Thick]},
 PlotLegend → {"Gammavert", "Lognormalvert"},
 LegendPosition → {-0.2, -0.3}, LegendShadow → None]
Clear[xmittel, xmax, xvari]
```

Out[350]=



■ Momente dieser speziellen "Gammaverteilung"

In[352]:=
```
xvari = 10;
xmittel = 6;
specialgammamomentvarmittel[n_] = Integrate[x^n * p1mod2[x], {x, 0, ∞},
 Assumptions → Re[n] ≥ 1]
Clear[xmittel, xmax, xvari]
```

Out[354]=

$$\frac{\left(\frac{5}{3}\right)^n \text{Gamma}\left[\frac{18}{5}+n\right]}{\text{Gamma}\left[\frac{18}{5}\right]}$$

■ Momente dieser speziellen Lognormalverteilung

In[356]:=
```
xvari = 10;
xmittel = 6;
speciallognormalmomentvarmittel[n_] = Integrate[x^n * p2mod2[x], {x, 0, ∞},
 Assumptions → Re[n] ≥ 1]
Clear[xmittel, xmax, xvari]
```

Out[358]=

$$3^{-\left(n+\frac{\text{Log}\left[\frac{648}{23}\right]}{\text{Log}\left[\frac{529}{324}\right]}\right)^2} \left(\frac{23}{2}\right)^{\frac{1}{2}\left(n+\frac{\text{Log}\left[\frac{648}{23}\right]}{\text{Log}\left[\frac{529}{324}\right]}\right)^2} e^{-\frac{\text{Log}\left[\frac{648}{23}\right]^2}{4\,\text{Log}\left[\frac{529}{324}\right]}}$$

die n-ten Momente aufgetragen über n

In[376]:=
```
gammamomentsvarmittel = Quiet[Table[N[specialgammamomentvarmittel[n]], {n, 1, 10}]]
lognormalmomentsvarmittel =
 Quiet[Table[N[speciallognormalmomentvarmittel[n]], {n, 1, 10}]]
ShowLegend[
 ListLogPlot[{gammamomentsvarmittel, lognormalmomentsvarmittel},
  PlotMarkers → Graphics[{PointSize[0.02], Point[{0, 0}]}], PlotStyle → {Red, Blue}],
 {Function[x, If[x == 0, Red, Blue]], 2, "Momente der Gammavert",
  "Momente der Lognormalvert", LegendPosition → {-0.7, 0.3},
  LegendPosition → {-0.1, 0.1}, LegendShadow → {0, 0},
  LegendSize → {1.2, 0.3}, LegendTextSpace → 15}
]
```

Out[376]= $\{6., 46., 429.333, 4722.67, 59\,820.4, 857\,426., 1.37188\times10^{7}, 2.42366\times10^{8}, 4.68574\times10^{9}, 9.84005\times10^{10}\}$

Out[377]= $\{6., 46., 450.63, 5640.75, 90\,221.4, 1.8439\times10^{6}, 4.81527\times10^{7}, 1.60679\times10^{9}, 6.85099\times10^{10}, 3.73252\times10^{12}\}$

Out[378]=



■ die Differenz der n-ten Momente aufgetragen über n

In[363]:=
```
momentdiffsvarmittel = Abs[gammamomentsvarmittel - lognormalmomentsvarmittel]
```

Out[363]= $\{7.10543\times10^{-15}, 7.10543\times10^{-15}, 21.2963, 918.085, 30\,401., 986\,474., 3.44339\times10^{7}, 1.36442\times10^{9}, 6.38241\times10^{10}, 3.63412\times10^{12}\}$

In[364]:=
```
ListLogPlot[momentdiffsvarmittel, PlotRange → {10^(-16), 10^15},
 Filling → Bottom, PlotStyle -> PointSize[Large]]
```

Out[364]=



gleicher Erwartungswert und Maximum an der gleichen Stelle

■ Verlauf

In[282]:=
```
xmittel = 6;
xmax = 5;
Plot[{p1mod1[x], p2mod1[x]}, {x, 0, 30},
 PlotStyle → {Directive[Black, Thick], Directive[Green, Thick]},
 PlotLegend → {"Gammavert", "Lognormalvert"},
 LegendPosition → {-0.2, -0.3}, LegendShadow → None]
Clear[xmittel, xmax, xvari]
```

Out[284]=



■ Momente dieser speziellen "Gammaverteilung"

In[365]:=
```
xmittel = 6;
xmax = 5;
specialgammamomentmaxmittel[n_] = Integrate[x^n * p1mod1[x], {x, 0, ∞},
 Assumptions → Re[n] ≥ 1]
Clear[xmittel, xmax, xvari]
```

Out[367]= $\frac{1}{120}$ Gamma[6 + n]

Momente dieser speziellen Lognormalverteilung

In[369]:=
```
xmittel = 6;
xmax = 5;
speciallognormalmomentmaxmittel [n_] = Integrate[x^n * p2mod1[x], {x, 0, ∞},
 Assumptions → Re[n] ≥ 1]
Clear[xmittel, xmax, xvari]
```

Out[371]= $5^{-\frac{1}{3}(-1+n)\,n}\,6^{\frac{1}{3}n\,(2+n)}$

■ die n-ten Momente aufgetragen über n

In[381]:=
```
gammamomentsmaxmittel = Quiet[Table[N[specialgammamomentmaxmittel [n]], {n, 1, 10}]]
lognormalmomentsmaxmittel =
 Quiet[Table[N[speciallognormalmomentmaxmittel [n]], {n, 1, 10}]]
ShowLegend[
 ListLogPlot[{gammamomentsmaxmittel, lognormalmomentsmaxmittel},
  PlotMarkers → Graphics[{PointSize[0.02], Point[{0, 0}]}], PlotStyle → {Red, Blue}],
 {Function[x, If[x == 0, Red, Blue]], 2, "Momente der Gammavert",
  "Momente der Lognormalvert", LegendPosition → {-0.7, 0.3},
  LegendPosition → {-0.1, 0.1}, LegendShadow → {0, 0},
  LegendSize → {1.2, 0.3}, LegendTextSpace → 15}
]
```

Out[381]= $\{6., 42., 336., 3024., 30\,240., 332\,640., 3.99168\times10^{6}, 5.18918\times10^{7}, 7.26486\times10^{8}, 1.08973\times10^{10}\}$

Out[382]= $\{6., 40.6528, 311.04, 2687.39, 26\,219.9, 288\,882., 3.59415\times10^{6}, 5.04963\times10^{7}, 8.01145\times10^{8}, 1.43532\times10^{10}\}$

Out[383]=



■ die Differenz der n-ten Momente aufgetragen über n

In[384]:=
```
momentdiffsmaxmittel = Quiet[
  Table[N[Abs[specialgammamomentmaxmittel [n] - speciallognormalmomentmaxmittel [n]]],
   {n, 1, 10}]]
```

Out[384]= $\{0., 1.34724, 24.96, 336.614, 4020.09, 43\,758.3, 397\,530., 1.3955\times10^{6}, 7.46593\times10^{7}, 3.45595\times10^{9}\}$

In[385]:=
```
ListLogPlot[momentdiffsmaxmittel, PlotRange → {10^(-16), 10^15},
  Filling → Bottom, PlotStyle -> PointSize[Large]]
```

Out[385]=